SYSML16-311 Revised Text

In the specification document for each of the stereotype definitions listed below, replace all the “Constraints” sub-clause by the ones given below:

ModelElements 
0. UML Extension
0. Stereotypes
Conform
Constraints
[bookmark: _GoBack83]1_general_is_viewpoint
The general classifier shall be an element stereotyped by Viewpoint
Viewpoint.allInstances()->exists(v | v.base_Class = self.base_Generalization.general)
[bookmark: _GoBack84]2_specific_is_view
The specific classifier shall be an element that is stereotyped by View
View.allInstances()->exists(v | v.base_Class = self.base_Generalization.specific)
0. Expose
Constraints
[bookmark: _GoBack85]1_client_is_view
The client shall be an element stereotyped by View.
self.base_Dependency.client->forAll(c | View.allInstances()->exists( v | v.base_Class = c))
0. View
Constraints
[bookmark: _GoBack86]1_single_viewpoint
A view shall only conform to a single viewpoint
Conform.allInstances()->select(base_Generalization.specific = self.base_Class)->size()  = 1
[bookmark: _GoBack87]2_viewpoint_derived_from_conform
The derived value of the viewpoint shall be the classifier stereotyped by Viewpoint that is the general classifier of the generalization relationship stereotyped by Conform for which the View is the specific classifier
self.viewpoint = Viewpoint.allInstances()->any(base_Class = Conform.allInstances()->any(base_Generalization.specific = self.base_Class).base_Generalization.general)
[bookmark: _GoBack88]3_stakeholder_derived_from_conform
The derived values of the stakeholder attribute shall be the names of the classifiers stereotyped by Stakeholder that are the values of the stakeholder attribute of the general classifier of the generalization relationship stereotyped by Conform for which the View is the specific classifier.
self.stakeholder = Viewpoint.allInstances()->any(base_Class = Conform.allInstances()->any(base_Generalization.specific = self.base_Class).base_Generalization.general).stakeholder
Viewpoint
Constraints
[bookmark: _GoBack89]1_method_derived_from_create_operations
The derived values of the method attribute shall be the names of the methods of the operations stereotyped by the UML Create stereotype on the classifier stereotyped by Viewpoint.
self.method = self.base_Class.allFeatures()->select(f | f.oclIsKindOf(UML::Operation))->select(o | Standard::Create.allInstances().base_BehavioralFeature->includes(o)).oclAsType(UML::Operation).method
[bookmark: _GoBack90]2_create_view_operation
The property ownedOperation shall include at least one operation named "View" with the UML Create stereotype applied.
self.base_Class.ownedOperation->exists(o | o.name=‘View’ and Standard::Create.allInstances().base_BehavioralFeature->includes(o))
Blocks 
0. UML Extension
Stereotypes
0. AdjunctProperty
Constraints
[bookmark: _GoBack19]1_principal_kind
The principal of an applied AdjunctProperty shall be a Connector, CallAction, ObjectNode, Variable, Parameter, submachine State, or InteractionUse.
self.principal.oclIsKindOf(UML::Connector) or self.principal.oclIsKindOf(UML::CallAction) or self.principal.oclIsKindOf(UML::ObjectNode) or self.principal.oclIsKindOf(UML::Variable) or self.principal.oclIsKindOf(UML::Parameter) or self.principal.oclIsKindOf(UML::InteractionUse) or (self.principal.oclIsKindOf(UML::State) and self.principal.oclAsType(UML::State).isSubmachineState)
[bookmark: _GoBack20]2_same_name
Properties to which AdjunctProperty applied shall have the same name as the principal.
self.principal.oclIsKindOf(UML::NamedElement) implies self.base_Property.name = self.principal.oclAsType(UML::NamedElement).name
[bookmark: _GoBack21]3_connector_and_callaction_composite
Properties with AdjunctProperty applied that have a Connector or CallAction as principal shall be composite.
self.principal.oclIsKindOf(UML::Connector) or self.principal.oclIsKindOf(UML::CallAction) implies self.base_Property.isComposite()
[bookmark: _GoBack22]4_same_owner
Properties with AdjunctProperty applied shall be owned by an element that owns the principal, at least indirectly, or one of that element’s specializations.
let owners: Set(UML::Element) = self.principal->closure(owner) in
let specializations: Set(UML::Element) = UML::Classifier.allInstances()->select(c | c->closure(general)->intersection(owners)->notEmpty()) in
owners->union(specializations)->includes(self.base_Property.owner)
[bookmark: _GoBack23]5_compatible_type
Properties with AdjunctProperty applied that have as principal a Connector, ObjectNode, Variable, or Parameter shall have the same type as the principal or one of that type's generalizations.
self.principal.oclIsKindOf(UML::Connector)
or self.principal.oclIsKindOf(UML::ObjectNode)
or self.principal.oclIsKindOf(UML::Variable)
or self.principal.oclIsKindOf(UML::Parameter)
implies  let principal_type: UML::Classifier = if self.principal.oclIsKindOf(UML::Connector) then
	self.principal.oclAsType(UML::Connector).type
else
	self.principal.oclAsType(UML::TypedElement).type.oclAsType(UML::Classifier)
endif in
principal_type->closure(general)->including(principal_type)->includes(self.base_Property.type)
[bookmark: _GoBack24]6_connector_principal_associationblock
Connectors that are principals of an applied AdjunctProperty shall have association blocks as types
self.principal.oclIsKindOf(UML::Connector) implies let type: UML::Association = self.principal.oclAsType(UML::Connector).type in Block.allInstances().base_Class->includes(type)
[bookmark: _GoBack25]7_adjunctproperty_connectorproperty_consistent
AdjunctProperty and ConnectorProperty applied to the same property shall have the same values for principal and connector, respectively.
AdjunctProperty.allInstances()->forAll(ap | let cp: ConnectorProperty = ConnectorProperty.allInstances()->any(base_Property=ap.base_Property) in (not cp.oclIsUndefined()) implies cp.connector = ap.principal)

[bookmark: _GoBack26]8_callAction_composite_and_consitent_type
Properties with AdjunctProperty applied that have a CallAction as principal shall be composite and be typed by the behavior invoked by the call action or one of that behavior's generalizations (for CallOperationActions, this shall generalize all behaviors that might be dispatched), and an upper multiplicity of one if the CallAction invokes a nonreentrant behavior.
self.principal.oclIsKindOf(UML::CallAction) implies
if self.principal.oclIsKindOf(UML::CallOperationAction) then
	let called: Set(UML::Behavior) = self.principal.oclAsType(UML::CallOperationAction).operation.method in
	if called->isEmpty() then
		self.base_Property.type->isEmpty()
	else
		self.base_Property.type->notEmpty() and called->forAll(b | b.general->including(b)->includes(self.base_Property.type))
	endif
else
	let called: UML::Behavior = if self.principal.oclIsKindOf(UML::CallBehaviorAction) then
		self.principal.oclAsType(UML::CallBehaviorAction).behavior
	else
		self.principal.oclAsType(UML::StartObjectBehaviorAction).behavior() endif in
	if called.oclIsUndefined() then
		self.base_Property.type.oclIsUndefined()
	else
		let behaviors: Set(UML::Behavior) = called->closure(generalization).oclAsType(UML::Behavior)->including(called)->asSet() in
		self.base_Property.type->notEmpty() and behaviors->includes(self.base_Property.type)
	endif
endif
[bookmark: _GoBack27]9_objectnode_multiplicity
Properties with AdjunctProperty applied that have an ObjectNode as principal shall have a lower multiplicity of zero and an upper multiplicity the same as or higher than the upperBound of the ObjectNode.
self.principal.oclIsKindOf(UML::ObjectNode) implies self.base_Property.lower = 0 and self.base_Property.upper >= self.principal.oclAsType(UML::ObjectNode).upperBound.unlimitedValue()
[bookmark: _GoBack17]10_multiplicity_same_or_less_restrictive
Properties with AdjunctProperty applied that have a Variable or Parameter as principal shall have a lower multiplicity the same as or lower than the lower multiplicity of their principal, and an upper multiplicity the same as or higher than the upper multiplicity of their principal
self.principal.oclIsKindOf(UML::MultiplicityElement) implies self.base_Property.lower <=  self.principal.oclAsType(UML::MultiplicityElement).lower and self.base_Property.upper >= self.principal.oclAsType(UML::MultiplicityElement).upper
[bookmark: _GoBack18]11_submachine_and_interactionuse_composite_and _compatible_type
Properties with AdjunctProperty applied that have an InteractionUse or submachine State as principal shall be composite and be typed by the interaction or state machine invoked by the interaction use or submachine State or one of their generalizations.
self.principal.oclIsKindOf(UML::InteractionUse) or self.principal.oclIsKindOf(UML::State)
implies
self.base_Property.isComposite
and
let behavior: UML::Behavior =  if self.principal.oclIsKindOf(UML::InteractionUse) then
	self.principal.oclAsType(UML::InteractionUse).refersTo
else
	self.principal.oclAsType(UML::State).submachine
endif in
if behavior.oclIsUndefined() then
	self.base_Property.type->isEmpty()
else
	self.base_Property.type->notEmpty()
	and behavior.oclAsType(UML::Classifier)->closure(general)->including(behavior)->includes(self.base_Property.type)
endif
BindingConnector
Constraints
[bookmark: _GoBack28]1_compatible_types
The two ends of a binding connector shall have either the same type or types that are compatible so that equality of their values can be defined.
self.base_Connector.end->at(1).role.type.conformsTo(self.base_Connector.end->at(2).role.type) or self.base_Connector.end->at(2).role.type.conformsTo(self.base_Connector.end->at(1).role.type)
Block
Constraints
[bookmark: _GoBack29]1_associations_binary
For an association in which both ends are typed by blocks, the number of ends shall be exactly two
UML::Association.allInstances()->select(a| a.memberEnd->forAll(e| e.type->notEmpty() and Block.allInstances().base_Class->includes(e.type)))->forAll(a | a.memberEnd->size()=2)
[bookmark: _GoBack30]2_connectors_binary
The number of ends of a connector owned by a block shall be exactly two. (In SysML, a binding connector is not typed by an association, so this constraint is not implied entirely by the preceding constraint.)
self.base_Class.ownedConnector->forAll(c | c.end->size()=2 )
[bookmark: _GoBack33]5_uml_connector_constraint_removed
The following constraint under 9.3.6, "Connector" in the UML 2 standard is removed by SysML: "[3] The ConnectableElements attached as roles to each ConnectorEnd owned by a Connector must [sic] be roles of the Classifier that owned the Connector, or they must be ports of such roles."
-- Cannot be expressed in OCL
[bookmark: _GoBack34]6_valueproperties_composite
If a property owned by a SysML Block or SysML ValueType is typed by a SysML ValueType, then the aggregation attribute of the property shall be "composite."
self.base_Class.ownedAttribute->select(a| ValueType.allInstances().base_DataType->includes(a.type))->forAll(a|a.isComposite())
[bookmark: _GoBack35]7_composition_acyclic
Within an instance of a SysML Block, the values of any property with composite aggregation (aggregation = composite) shall not contain the block in any of its own properties that also have composite aggregation, or within any unbroken chain of properties that all have composite aggregation. (Within an instance of a SysML Block, the instances of properties with composite aggregation shall form an acyclic graph.)
self.base_Class->closure(part->select(p|p.type.oclIsKindOf(UML::Class)).type.oclAsType(UML::Class))->excludes(self.base_Class)
[bookmark: _GoBack36]8_specializations_are_blocks
UML::Classifier.allInstances()->select(c | c.general->includes(self.base_Class))->forAll(c | Block.allInstances()->includes(c))
[bookmark: _GoBack37]9_uml_connectorend_constraint_removed
The following constraint under 9.3.7, "ConnectorEnd" in the UML 2 standard is removed by SysML: "[3] The property held in self.partWithPort must [sic] not be a Port."
--Cannot be expressed in OCL

BoundReference
Constraints
[bookmark: _GoBack38]1_bindingconnector_end
Properties to which BoundReference is applied shall be the role of a connector end of at least one binding connector, or generalized by such a property through redefinition
BindingConnector.allInstances().base_Connector.end.role->exists(r | r=self.base_Property or self.base_Property->closure(redefinedElement)->includes(r))
[bookmark: _GoBack39]2_opposite_bindingconnector_end
The value of boundEnd shall be a connector end of a binding connector, as identified in constraint 1, opposite the property, as identified in constraint 1_bindingconnector_end.
let opposite: UML::ConnectorEnd = BindingConnector.allInstances().base_Connector.end->any(e | e.role=self.base_Property or self.base_Property->closure(redefinedElement)->includes(e.role)) in
self.boundEnd = opposite.owner.oclAsType(UML::Connector).end->any(e | e<>opposite)
[bookmark: _GoBack40]3_navigable
The role of boundEnd shall be a property accessible by navigation from instances of the block owning the property to which BoundReference is applied, but shall not be the property to which BoundReference is applied, or one that it is related to by redefinition.
self.base_Property.association->notEmpty() and
self.boundEnd.definingEnd->notEmpty() and
self.base_Property.association.navigableOwnedEnd->includes(self.boundEnd.definingEnd)
[bookmark: _GoBack41]4_propertypath_consistency
The last value of bindingPath shall be the role of boundEnd, and the other values shall be the propertyPath of the NestedConnectorEnd applied to boundEnd, if any.
self.boundEnd.definingEnd = self.bindingPath->last() and
(let nce: NestedConnectorEnd = NestedConnectorEnd.allInstances()->any(n| n.base_ConnectorEnd=self.boundEnd) in
nce->oclIsUndefined() or self.bindingPath->subSequence(1, self.bindingPath->size()-1) = nce.propertyPath)
[bookmark: _GoBack42]5_reference_or_valueproperty
Properties to which BoundReference is applied shall either be reference properties or value properties.
ValueType.allInstances().base_DataType->includes(self.base_Property.type) or not self.base_Property.isComposite()
[bookmark: _GoBack43]6_ordered_nonunique
Properties with BoundReference applied that have an upper multiplicity greater than one shall be ordered and non-unique.
self.base_Property.upper > 1 implies self.base_Property.isOrdered and not self.base_Property.isUnique
[bookmark: _GoBack44]7_cannot_redefine_boundreference
BoundReferences shall not be applied to properties that are related by redefinition to other properties with BoundReference applied.
self.base_Property.redefinedElement->notEmpty() implies BoundReference.allInstances().base_Property->excludesAll(self.base_Property.redefinedElement)
[bookmark: _GoBack45]8_notbounded_to_itslef
The binding connector identified in constraint  1_bindingconnector_end shall not have the same property on both ends, or properties related by redefinition.
let e1: UML::ConnectorEnd = self.boundEnd.owner.oclAsType(UML::Connector).end->at(1) in
let e2: UML::ConnectorEnd = self.boundEnd.owner.oclAsType(UML::Connector).end->at(2) in
e1.role <> e2.role and
(e1.role.oclIsKindOf(UML::Property) and e2.role.oclIsKindOf(UML::Property) implies e1.role.oclAsType(UML::Property).redefinedElement->excludes(e2.role) and e2.role.oclAsType(UML::Property).redefinedElement->excludes(e1.role))
ClassifierBehaviorProperty
Constraints
[bookmark: _GoBack46]1_owner_classifierbehavior
ClassifierBehaviorProperty shall only be applied to properties owned (not inherited) by blocks that have classifier behaviors.
Block.allInstances().base_Class->exists(c | c.ownedAttribute->includes(self.base_Property) and c.classifierBehavior->notEmpty())
[bookmark: _GoBack47]2_composite
Properties to which ClassifierBehaviorProperty is applied shall be composite
self.base_Property.isComposite
[bookmark: _GoBack48]3_typed_by_classifierbehavior
Properties to which ClassifierBehaviorProperty applied shall be typed by the classifier behavior of their owning block or a generalization of the classifier behavior.
let clBehavior: UML::Behavior = self.base_Property.owner.oclAsType(UML::Class).classifierBehavior in
self.base_Property.type->notEmpty() and clBehavior->closure(general)->including(clBehavior)->includes(self.base_Property.type)
ConnectorProperty
Constraints
[bookmark: _GoBack49]1_block_property
ConnectorProperty shall only be applied to properties of classes stereotyped by Block.
Block.allInstances().base_Class->exists(c | c.ownedAttribute->includes(self.base_Property))
[bookmark: _GoBack50]2_owned_or_inherited
The connector attribute of the applied stereotype shall refer to a connector owned or inherited by a block owning the property on which the stereotype is applied.
let owner: UML::Class = Block.allInstances().base_Class->any(c | c.ownedAttribute->includes(self.base_Property)) in
owner->closure(general)->select(oclIsKindOf(UML::Class)).oclAsType(UML::Class).ownedConnector->flatten()->includes(self.connector)
[bookmark: _GoBack51]3_composite
The aggregation of a property stereotyped by ConnectorProperty shall be composite.
self.base_Property.isComposite
[bookmark: _GoBack52]4_typed_by_associationblock
The type of the connector referred to by a connector attribute shall be an association class stereotyped by Block.
Block.allInstances().base_Class->exists(c | c.oclIsKindOf(UML::AssociationClass) and self.connector.type = c)
[bookmark: _GoBack53]5_same_name_and_type
A property stereotyped by ConnectorProperty shall have the same name and type as the connector referred to by the connector attribute.
self.base_Property.name = self.connector.name and self.base_Property.type = self.connector.type
DirectedRelationshipPropertyPath
Constraints
[bookmark: _GoBack54]1_sourcecontext_iif_property
sourceContext shall have a value when source is a property, otherwise it shall not have a value
self.base_DirectedRelationship.source->exists(s | s.oclIsKindOf(UML::Property)) xor self.sourceContext->isEmpty()

[bookmark: _GoBack55]2_targetcontext_iif_property
targetContext shall have a value when target is a property, otherwise it shall not have a value.
self.base_DirectedRelationship.source->exists(s | s.oclIsKindOf(UML::Property)) xor self.sourceContext->isEmpty()

[bookmark: _GoBack56]3_sourcepropertypath_implies_property
source shall be a property when sourcePropertyPath has a value.
self.sourcePropertyPath->notEmpty() implies self.base_DirectedRelationship.source->forAll(s | s.oclIsKindOf(UML::Property))

[bookmark: _GoBack57]4_targetpropertypath_implies_property
target shall be a property when targetPropertyPath has a value.
self.targetPropertyPath->notEmpty() implies self.base_DirectedRelationship.target->forAll(s | s.oclIsKindOf(UML::Property))

[bookmark: _GoBack58]5_sourcecontext_owns_sourcepath_first
The property in the first position of the sourcePropertyPath list, if any, shall be owned by the sourceContext or one of its generalizations
self.sourcePropertyPath->notEmpty() implies self.sourceContext.allAttributes()->includes(self.sourcePropertyPath->first())
[bookmark: _GoBack59]6_targetcontext_owns_targetpath_first
The property in the first position of the targetPropertyPath list, if any, shall be owned by the targetContext or one of its generalizations.
self.targetPropertyPath->notEmpty() implies self.targetContext.allAttributes()->includes(self.targetPropertyPath->first())
[bookmark: _GoBack60]7_path_and_owners_consistency
The property at each successive position of the sourcePropertyPath and targetPropertyPath, following the first position, shall be owned by the Block or ValueType that types the property at the immediately preceding position, or a generalization of the Block or ValueType.
let BlocksAndValueTypes: Set(UML::Classifier) = Block.allInstances().base_Class->union(ValueType.allInstances().base_DataType)->asSet() in
(self.sourcePropertyPath->size() >1 implies self.sourcePropertyPath->subSequence(2, self.sourcePropertyPath->size())->forAll(p |
	let pp: UML::Property = self.sourcePropertyPath->at(self.sourcePropertyPath->indexOf(p)-1) in
	let owners: Set(UML::Classifier) = pp.type.oclAsType(UML::Classifier)->select(t | BlocksAndValueTypes->includes(t))->closure(general)->including(pp.type.oclAsType(UML::Classifier)) in
	owners->includes(p.owner)))
and
(self.targetPropertyPath->size() >1 implies self.targetPropertyPath->subSequence(2, self.targetPropertyPath->size())->forAll(p |
	let pp: UML::Property = self.targetPropertyPath->at(self.targetPropertyPath->indexOf(p)-1) in
	let owners: Set(UML::Classifier) = pp.type.oclAsType(UML::Classifier)->select(t | BlocksAndValueTypes->includes(t))->closure(general)->including(pp.type.oclAsType(UML::Classifier)) in
	owners->includes(p.owner)))
[bookmark: _GoBack61]8_sourcepath_last_type_owns_source
The type of the property at the last position of the sourcePropertyPath list shall own or inherit the source of the stereotyped directed relationship.
self.sourcePropertyPath->notEmpty() implies self.sourcePropertyPath->last().type.oclAsType(UML::Classifier).allAttributes()->includesAll(self.base_DirectedRelationship.source)
[bookmark: _GoBack62]9_targetpath_last_type_owns_target
The type of the property at the last position of the targetPropertyPath list shall own or inherit the target of the stereotyped directed relationship.
self.targetPropertyPath->notEmpty() implies self.targetPropertyPath->last().type.oclAsType(UML::Classifier).allAttributes()->includesAll(self.base_DirectedRelationship.target)
DistributedProperty
Constraints
[bookmark: _GoBack63]1_block_or_valuetype
The DistributedProperty stereotype shall only be applied to properties of classifiers stereotyped by Block or ValueType.
Block.allInstances().base_Class.oclAsType(UML::Classifier)->union(ValueType.allInstances().base_DataType)->includes(self.base_Property.owner)

ElementPropertyPath
Constraints
[bookmark: _GoBack64]1_path_consistency
The property at each successive position of the propertyPath attribute, following the first position, shall be owned by the Block or ValueType that types the property at the immediately preceding position, or a generalization of the Block or ValueType.
let BlocksAndValueTypes: Set(UML::Classifier) = Block.allInstances().base_Class->union(ValueType.allInstances().base_DataType)->asSet() in
self.propertyPath->size() >1 implies self.propertyPath->subSequence(2, self.propertyPath->size())->forAll(p |
let pp: UML::Property = self.propertyPath->at(self.propertyPath->indexOf(p)-1) in
let owners: Set(UML::Classifier) = pp.type.oclAsType(UML::Classifier)->select(t| BlocksAndValueTypes->includes(t))->including(pp.type.oclAsType(UML::Classifier)) in
owners->includes(p.owner))
EndPathMultiplicity
Constraints
[bookmark: _GoBack65]1_redefinition
Properties to which EndPathMultiplicity is applied shall be related by redefinition to a property to which BoundReference is applied.
self.base_Property.redefinedProperty->notEmpty() and BoundReference.allInstances().base_Property->exists(p | self.base_Property.redefinedProperty->includes(p))
[bookmark: _GoBack66]2_non_negative
endPathLower shall be non-negative.
self.lower >= 0
NestedConnectorEnd
Constraints
[bookmark: _GoBack67]1_propertypath_first_owned_by_connector_owner
The first property in propertyPath shall be owned by the block that owns the connector, or one of the block's generalizations.
let owningBlock: UML::Class = self.base_ConnectorEnd.owner.oclAsType(UML::Connector).owner.oclAsType(UML::Class) in
(not owningBlock.oclIsUndefined()) and owningBlock->closure(general)->including(owningBlock)->includes(self.propertyPath->first().owner)
[bookmark: _GoBack68]2_propertypath_last_type_owns_role
The type of the property at the last position of the propertyPath list shall own or inherit the role property of the stereotyped connector end
let type: UML::Classifier = self.propertyPath->last().type.oclAsType(UML::Classifier) in
(not type.oclIsUndefined()) and type.allFeatures()->includes(self.base_ConnectorEnd.role)

[bookmark: _GoBack69]ParticipantProperty
Constraints
[bookmark: _GoBack70]1_associationblock
ParticipantProperty shall only be applied to properties of association classes stereotyped by Block.
self.base_Property.class.oclIsKindOf(UML::AssociationClass) and  Block.allInstances().base_Class->includes(self.base_Property.class)
[bookmark: _GoBack71]2_memberend
ParticipantProperty shall not be applied to properties that are member ends of an association.
UML::Association.allInstances().memberEnd->flatten()->excludes(self.base_Property)
[bookmark: _GoBack72]3_aggregationkind_none
The aggregation of a property stereotyped by ParticipantProperty shall be none.
self.base_Property.aggregation = UML::AggregationKind::none
[bookmark: _GoBack73]4_end_owner
The end attribute of the applied stereotype shall refer to a member end of the association block owning the property on which the stereotype is applied.
self.base_Property.association.memberEnd->includes(self.end)
[bookmark: _GoBack74]5_same_type
A property stereotyped by ParticipantProperty shall have the same type as the property referred to by the end attribute.
self.base_Property.type = self.end.type
[bookmark: _GoBack75]6_multiplicity_1
A property to which the ParticipantProperty is applied shall have a multiplicity of 1.
self.base_Property.lower = 1 and self.base_Property.upper = 1
PropertySpecificType
Constraints
[bookmark: _GoBack76]1_only_one_property
A classifier to which the PropertySpecificType stereotype is applied shall be referenced as the type of one and only one property.
UML::Property.allInstances()->select(p | p.type = self.base_Classifier)->size() = 1
[bookmark: _GoBack77]2_classifer_unnamed
The name of a classifier to which a PropertySpecificType is applied shall be missing. (The "name" attribute of the NamedElement metaclass shall be empty.)
self.base_Classifier.name->isEmpty()
ValueType
Constraints
[bookmark: _GoBack78]1_specializations_are_valuetypes
Any classifier that specializes a ValueType shall also have the ValueType stereotype applied.
UML::Classifier.allInstances()->forAll(c | c.general->includes(self.base_DataType) implies ValueType.allInstances().base_DataType->includes(c))
[bookmark: _GoBack79]2_unit
The unit of a ValueType, if any, shall be an InstanceSpecification classified by SysML's Unit block in the UnitAndQuantityKind model library or a specialization of it.
self.unit->notEmpty() and self.unit.classifier->notEmpty() implies self.unit.classifier->forAll(c | c.oclIsKindOf(Libraries::UnitAndQuantityKind::Unit))
[bookmark: _GoBack80]3_quantitykind
The quantityKind of a ValueType, if any, shall be an InstanceSpecification classified by SysML's QuantityKind block in the UnitAndQuantityKind model library or a specialization of it.
self.quantityKind->notEmpty() and self.quantityKind.classifier->notEmpty() implies self.quantityKind.classifier->forAll(c | c.oclIsKindOf(Libraries::UnitAndQuantityKind::QuantityKind))
Ports&Flows 
0. UML Extension
2. Stereotypes
AcceptChangeStructuralFeatureEventAction
Constraints
[bookmark: _GoBack91]1_one_trigger
The action has exactly one trigger, the event of which shall be a change structural feature event.
self.base_AcceptEventAction.trigger->size()=1 and let trigger: UML::Trigger = self.base_AcceptEventAction.trigger->any(true) in
ChangeStructuralFeatureEvent.allInstances().base_ChangeEvent->includes(trigger.event)
[bookmark: _GoBack92]2_two_resultpins
The action has two result pins with type and ordering the same as the type and ordering of the structural feature of the trigger event, and multiplicity compatible with the multiplicity of the structural feature.
let event: ChangeStructuralFeatureEvent = ChangeStructuralFeatureEvent.allInstances()->any(e | e.base_ChangeEvent = self.base_AcceptEventAction.trigger->any(true).event) in
self.base_AcceptEventAction.result->size() = 2 and
self.base_AcceptEventAction.result->forAll(r | r.type = event.structuralFeature.type
	and r.isOrdered = event.structuralFeature.isOrdered
	and r.lower <= event.structuralFeature.lower
	and r.upper >= event.structuralFeature.upper)

[bookmark: _GoBack93]3_context_owns_structuralfeature
The structural feature of the trigger event shall be owned by or inherited by the context of the behavior containing the action. (The context of a behavior is either its owning block or itself if it is not owned by a block. See definition in the UML 2 standard.)
let event: ChangeStructuralFeatureEvent = ChangeStructuralFeatureEvent.allInstances()->any(e | e.base_ChangeEvent = self.base_AcceptEventAction.trigger->any(true).event) in
self.base_AcceptEventAction._’context’->notEmpty() and self.base_AcceptEventAction._’context’.allFeatures()->includes(event.structuralFeature)
[bookmark: _GoBack94]4_can_access_structuralfeature
Visibility of the structural feature of the trigger event shall allow access to the object performing the action.
let event: ChangeStructuralFeatureEvent = ChangeStructuralFeatureEvent.allInstances()->any(e | e.base_ChangeEvent = self.base_AcceptEventAction.trigger->any(true).event) in

if event.structuralFeature.visibility = UML::VisibilityKind::private then
	self.base_AcceptEventAction._’context’.feature->includes(event.structuralFeature)
	
else if event.structuralFeature.visibility = UML::VisibilityKind::protected then
	self.base_AcceptEventAction._’context’.allFeatures()->includes(event.structuralFeature)
	
else if event.structuralFeature.visibility = UML::VisibilityKind::_’package’ then
	let thePackage: UML::Package = event.structuralFeature.allNamespaces()->select(n | n.oclIsKindOf(UML::Package))->first().oclAsType(UML::Package) in	
	(not thePackage.oclIsUndefined()) and (
		let index: Integer = event.structuralFeature.allNamespaces()->indexOf(thePackage) in
		event.structuralFeature.allNamespaces()->subOrderedSet(1, index)
		->iterate(n; acc: Boolean=true |
			acc and not (n.visibility=UML::VisibilityKind::private or n.visibility=UML::VisibilityKind::protected))	)
else
	true
endif endif endif
[bookmark: _GoBack95]5_uml_constraint_removed
The constraint under 11.3.2, "AcceptEventAction" in the UML 2 standard, "[2] There are no output pins if the trigger events are only ChangeEvents," shall be removed for accept event actions that have
AcceptChangeStructuralFeatureEventAction applied.
-- cannot be expressed in OCL

0. ChangeStructuralFeatureEvent
Constraints
[bookmark: _GoBack96]1_not_static
The structural feature shall not be static
not self.structuralFeature.isStatic
[bookmark: _GoBack97]2_one_featuringclassifier
The structural feature shall have exactly one featuringClassifier
self.structuralFeature.featuringClassifier->size()=1
DirectedFeature
Constraints
[bookmark: _GoBack98]1_behavioralfeature_or_not_flowproperty
DirectedFeature shall only be applied to behavioral features, or to properties that do not have FlowProperty applied, including on subsetted or redefined features.
self.base_Feature.oclIsKindOf(UML::BehavioralFeature) or
(self.base_Feature.oclIsKindOf(UML::Property) and
	let property: UML::Property = self.base_Feature.oclAsType(UML::Property) in
	FlowProperty.allInstances().base_Property->excludesAll(property.redefinedProperty->union(property.subsettedProperty)->including(property)))
[bookmark: _GoBack99]2_method_if_provided
A non-provided operation shall not be associated with a behavior as its method.
self.base_Feature.oclIsKindOf(UML::Operation) and self.featureDirection=FeatureDirection::required implies self.base_Feature.oclAsType(UML::Operation).method->isEmpty()
0. FlowProperty
Constraints
[bookmark: _GoBack100]1_restricted_types
A FlowProperty shall be typed by a ValueType, Block, or Signal.
Block.allInstances().base_Class->includes(self.base_Property.type)
or ValueType.allInstances().base_DataType->includes(self.base_Property.type)
or self.base_Property.oclIsKindOf(UML::Signal)
FullPort
Constraints
[bookmark: _GoBack101]1_not_proxy
Full ports shall not also be proxy ports. This applies even if some of the stereotypes are on subsetted or redefined ports.
ProxyPort.allInstances()->excludes(self.base_Port)
[bookmark: _GoBack102]2_not_bound_to_fullport
Binding connectors shall not link full ports (either directly or indirectly through other binding connectors) to other composite properties of the block owning the full port (or that block's generalizations or specializations), unless the composite properties are non-full ports.
let fullPorts: Set(UML::Port) = FullPort.allInstances().base_Port->asSet() in
BindingConnector.allInstances().base_Connector->select(c | c.end.role->includes(self.base_Port))->forAll(c | fullPorts->excludesAll(c.end.role->reject(r | r=self.base_Port)))
[bookmark: _GoBack103]3_not_behavioral
Full ports shall not be behavioral (isBehavior=false).
not self.base_Port.isBehavior
[bookmark: _GoBack104]4_not_conjugated
Full ports shall not be conjugated (isConjugated=false).
not self.base_Port.isConjugated
InterfaceBlock
Constraints
[bookmark: _GoBack105]1_no_behavior
Interface blocks shall not own or inherit behaviors, have classifier behaviors, or methods for their behavioral features.
self.base_Class.member->select(m | m.oclIsKindOf(UML::Behavior))->isEmpty()
and self.base_Class.member->select(m | m.oclIsKindOf(UML::Operation)).oclAsType(UML::Operation).method->flatten()->isEmpty()
[bookmark: _GoBack106]2_no_part
Interface blocks' composite properties are either ports, value properties or flow properties
self.base_Class.ownedAttribute->select(a|a.isComposite)->forAll(a | a.oclIsKindOf(UML::Port)
or ValueType.allInstances().base_DataType->includes(a))
[bookmark: _GoBack107]3_interfaceblock_typed_ports
Ports owned by interface blocks shall only be typed by interface blocks.
self.base_Class.ownedPort->forAll(p|InterfaceBlock.allInstances().base_Class->includes(p.type))
InvocationOnNestedPortAction
Constraints
[bookmark: _GoBack108]1_onPort_defined
The onPort property of an invocation action shall have a value when this stereotype is applied.
self.base_InvocationAction.onPort->notEmpty()
[bookmark: _GoBack109]2_onnestedport_first_owned_by_target_type
The port at the first position in the onNestedPort list shall be owned (directly or via inheritance) by a block that types the target pin of the invocation action, or one of the block's generalizations.
let target: UML::InputPin = if self.base_InvocationAction.oclIsKindOf(UML::CallOperationAction) then
	self.base_InvocationAction.oclAsType(UML::CallOperationAction).target
else if self.base_InvocationAction.oclIsKindOf(UML::SendSignalAction) then
	self.base_InvocationAction.oclAsType(UML::SendSignalAction).target
else if self.base_InvocationAction.oclIsKindOf(UML::SendObjectAction) then
	self.base_InvocationAction.oclAsType(UML::SendObjectAction).target
else
	invalid
endif endif endif in
not target.oclIsUndefined() and (
let target_type: UML::Class = Block.allInstances()->any(b | b.base_Class = target.type).base_Class in
not target_type.oclIsUndefined() and target_type.allFeatures()->includes(self.onNestedPort->first()))


[bookmark: _GoBack110]3_path_consistency
The first constraint of ElementPropertyPath shall apply to onNestedPort.
self.onNestedPort->size() >1 implies self.propertyPath->subSequence(2, self.onNestedPort->size())->forAll(p |
let pp: UML::Property = self.onNestedPort->at(self.onNestedPort->indexOf(p)-1) in
let owners: Set(UML::Classifier) = pp.type.oclAsType(UML::Classifier)->including(pp.type.oclAsType(UML::Classifier)) in
owners->includes(p.owner))
[bookmark: _GoBack111]4_onnestedport_last_type_owns_invocation_onPort
The type of the port at the last position of the onNestedPort list shall own or inherit the onPort port of the stereotyped invocation action.
self.onNestedPort->last().type.oclAsType(UML::Classifier).allFeatures()->includes(self.base_InvocationAction.onPort)
ItemFlow
Constraints
[bookmark: _GoBack113]1_source_and_target_linked
A Connector or an Association, or an inherited Association shall exist between the source and the target of the
InformationFlow.
let target: UML::NamedElement = self.base_InformationFlow.informationTarget->any(true) in
let targets: Set(UML::NamedElement) = if target.oclIsKindOf(UML::Classifier) then
	target.oclAsType(UML::Classifier)->closure(general)->including(target)
else
	target->asSet()
endif in
let source: UML::NamedElement = self.base_InformationFlow.informationSource->any(true) in
let sources: Set(UML::NamedElement) = if source.oclIsKindOf(UML::Classifier) then
	source.oclAsType(UML::Classifier)->closure(general)->including(source)
else
	source->asSet()
endif in
UML::Association.allInstances()->exists(a | a.memberEnd->intersection(targets)->notEmpty() and  a.memberEnd->intersection(sources)->notEmpty())
or UML::Connector.allInstances()->exists(c | c.end->intersection(targets)->notEmpty() and  c.end->intersection(sources)->notEmpty())


[bookmark: _GoBack114]2_type_restricted
An ItemFlow itemProperty shall be typed by a ValueType, Block, or Signal.
ValueType.allInstances().base_DataType->includes(self.itemProperty.type)
or Block.allInstances().base_Class->includes(self.itemProperty.type)
or UML::Signal.allInstances()->includes(self.itemProperty.type)
[bookmark: _GoBack115]3_itemproperty_common_owner
If itemProperty has a value it shall be a property of the common (possibly indirect) owner of the source and the target.
self.itemProperty->notEmpty() implies
(let target: UML::Element = self.base_InformationFlow.informationTarget->any(true) in
let source: UML::Element = self.base_InformationFlow.informationSource->any(true) in
target.oclIsKindOf(UML::Property) and
source.oclIsKindOf(UML::Property) and
let owners: Set(UML::Classifier) = target->closure(owner)->select(o1 | o1.oclIsKindOf(UML::Classifier))->asSet()
->intersection(source->closure(owner)->select(o2 | o2.oclIsKindOf(UML::Classifier))).oclAsType(UML::Classifier)->asSet() in
owners.attribute->flatten()->includes(self.itemProperty))
[bookmark: _GoBack116]4_association_xor_itemproperty
itemProperty shall not have a value if the item flow is realized by an Association.
self.base_InformationFlow.realization->exists(r | r.oclIsKindOf(UML::Association)) implies self.itemProperty->isEmpty()
[bookmark: _GoBack117]5_same_type
If an ItemFlow has an itemProperty, one of the classifiers of conveyed items shall be the same as the type of the item
property.
self.itemProperty->notEmpty() implies self.base_InformationFlow.conveyed->includes(self.itemProperty.type)
[bookmark: _GoBack118]6_same_name
If an ItemFlow has an itemProperty, its name shall be the same as the name of the item flow.
self.itemProperty->notEmpty() implies self.itemProperty.name = self.base_InformationFlow.name
ProxyPort
Constraints
[bookmark: _GoBack119]1_not_fullport
Proxy ports shall not also be full ports. This applies even if some of the stereotypes are on subsetted or redefined ports.
FullPort.allInstances()->excludes(self.base_Port)
[bookmark: _GoBack120]2_interfaceblock
Proxy ports shall only be typed by interface blocks.
InterfaceBlock.allInstances().base_Class->includes(self.base_Port.type)
[bookmark: _GoBack121]3_subports_are_proxyports
Ports owned by the type of a proxy port shall be proxy ports.
ProxyPort.allInstances().base_Port->includesAll(self.base_Port.class.ownedPort)
TriggerOnNestedPort
Constraints
[bookmark: _GoBack122]1_single_proxyport
The port property of the stereotyped trigger shall have exactly one value, and the value cannot be a full port.
self.base_Trigger.port->size()=1 and FullPort.allInstances().base_Port->excludes(self.base_Trigger.port)
[bookmark: _GoBack123]2_no_fullport
The values of the onNestedPort property shall not be full ports.
FullPort.allInstances().base_Port->excludesAll(self.onNestedPort)
[bookmark: _GoBack124]3_onnestedport_first_owned_by_context
The port at the first position in the onNestedPort list shall be owned by a block in which the trigger is used, or one of the block's generalizations.
let theContext: UML::Classifier = if self.base_Trigger.owner.oclIsKindOf(UML::Action) then
	self.base_Trigger.owner.oclAsType(UML::Action)._’context’.oclAsType(UML::Class)
else
	self.base_Trigger.owner.oclAsType(UML::Transition).containingStateMachine()._’context’.oclAsType(UML::Class)
endif in
let owners: Set(UML::Classifier) = theContext->closure(general)->including(theContext) in
owners->includes(self.onNestedPort->first().owner)
[bookmark: _GoBack125]4_path_consistency
The first constraint of ElementPropertyPath shall apply to onNestedPort.
self.onNestedPort->size() >1 implies self.onNestedPort->subSequence(2, self.onNestedPort->size())->forAll(p |
let np: UML::Port = self.onNestedPort->at(self.onNestedPort->indexOf(p)-1) in
let owners: Set(UML::Classifier) = np.type.oclAsType(UML::Classifier)->including(np.type.oclAsType(UML::Classifier)) in
owners->includes(p.owner))
[bookmark: _GoBack126]5_onnestedport_last_type_owns_trigger_port
The type of the port at the last position of the onNestedPort list must own or inherit the port of the stereotyped trigger.
self.onNestedPort->last().type.oclAsType(UML::Classifier).allFeatures()->includes(self.base_Trigger.port)
ConstraintBlocks 
0. UML Extension
3. Stereotypes
ConstraintBlock
Constraints
[bookmark: _GoBack81]1_constraintparameters_only
A constraint block shall not own any structural or behavioral elements beyond the properties that define its constraint parameters, constraint properties that hold internal usages of constraint blocks, binding connectors between its internally nested constraint parameters, constraint expressions that define an interpretation for the constraint block, and general-purpose model management and crosscutting elements.
-- Cannot be expressed in OCL
2_specializations_are_constraintblocks
Any classifier that specializes a ConstraintBlock shall also have the ConstraintBlock stereotype applied.
UML::Classifier.allInstances()->forAll(c | c.general->includes(self.base_Class) implies ConstraintBlock.allInstances().base_Class->includes(c))
[bookmark: _GoBack82]3_composite
Any property of a block that is typed by a ConstraintBlock shall have composite aggregation.
self.base_Class.ownedAttribute->forAll(p| p.isComposite)

Activities 
0. UML Extension
4. Stereotypes
0. ControlOperator
Constraints
1_one_parameter_controlvalue
When the «controlOperator» stereotype is applied, the behavior or operation shall have at least one parameter typed by ControlValue. If the stereotype is not applied, the behavior or operation may not have any parameter typed by ControlValue.
self.base_Behavior->notEmpty() implies self.base_Behavior.ownedParameter->exists(p | p.type=SysML::Libraries::ControlValues::ControlValue)
and
self.base_Operation->notEmpty() implies self.base_Operation.ownedParameter->exists(p | p.type=SysML::Libraries::ControlValues::ControlValue)

[bookmark: _GoBack1]2_controloperator_operation_method
A behavior shall have the «controlOperator» stereotype applied if it is a method of an operation that has the «controlOperator» stereotype applied.
(self.base_Operation->notEmpty() and self.base_Operation.method->notEmpty()) implies  self.base_Operation.method->forAll(b | ControlOperator.allInstances().base_Behavior->includes(b))
Discrete
Constraints
[bookmark: _GoBack2]1_not_continuous
The «discrete» and «continuous» stereotypes shall not be applied to the same element at the same time.
(self.base_ActivityEdge->notEmpty() implies Continuous.allInstances().base_ActivityEdge->excludes(self.base_ActivityEdge))
and
(self.base_Parameter->notEmpty() implies Continuous.allInstances().base_Parameter->excludes(self.base_Parameter))

NoBuffer
Constraints
[bookmark: _GoBack3]1_not_overwrite
The «nobuffer» and «overwrite» stereotypes cannot be applied to the same element at the same time.
Overwrite.allInstances().base_ObjectNode->excludes(self.base_ObjectNode)
Optional
Constraints
[bookmark: _GoBack4]1_lower_is_0
A parameter with the «optional» stereotypes applied shall have multiplicity.lower equal to zero, otherwise multiplicity.lower shall be greater than zero
UML::Parameter.allInstances()->forAll(p | Optional.allInstances().base_Parameter->includes(p) xor p.lower > 0)
Overwrite
Constraints
[bookmark: _GoBack5]1_not_nobuffer
The «overwrite» and «nobuffer» stereotypes cannot be applied to the same element at the same time.
NoBuffer.allInstances().base_ObjectNode->excludes(self.base_ObjectNode)
Probability
Constraints
[bookmark: _GoBack6]1_source_decisionnode_or_objectnode
The «probability» stereotype shall only be applied to activity edges that have decision nodes or object nodes as sources, or to output parameter sets.
(self.base_ActivityEdge->notEmpty() implies self.base_ActivityEdge.source.oclIsKindOf(UML::DecisionNode))
and
(self.base_ParameterSet->notEmpty() implies self.base_ParameterSet.parameter->forAll(p | p.direction=UML::ParameterDirectionKind::out))

[bookmark: _GoBack7]2_all_outgoing_edges
When the «probability» stereotype is applied to an activity edge, then it shall be applied to all edges coming out of the same source.
self.base_ActivityEdge->notEmpty() implies Probability.allInstances().base_ActivityEdge->includesAll(self.base_ActivityEdge.target.incoming)

[bookmark: _GoBack8]3_all_parametersets
When the «probability» stereotype is applied to an output parameter set, it shall be applied to all the parameter sets of the behavior or operation owning the original parameter set.
self.base_ParameterSet->notEmpty() implies Probability.allInstances().base_ParameterSet->includesAll(self.base_ParameterSet.namespace.ownedMember->select(m | m.oclIsKindOf(UML::ParameterSet)))

[bookmark: _GoBack9]4_all_outputparameter_in_parametersets
When the «probability» stereotype is applied to an output parameter set, all the output parameters shall be in some
parameter set.
self.base_ParameterSet->notEmpty() and self.base_ParameterSet.parameter->forAll(p | p.direction=UML::ParameterDirectionKind::out)
implies self.base_ParameterSet.owner.ownedElement->excludesAll(UML::Parameter.allInstances()->select(p | p.direction=UML::ParameterDirectionKind::out))
Rate
Constraints
[bookmark: _GoBack10]1_streaming
When the «rate» stereotype is applied to a parameter, the parameter shall be streaming.
self.base_Parameter->notEmpty() implies self.base_Parameter.isStream
[bookmark: _GoBack11]2_edges_rates
The rate of a parameter shall be less than or equal to rates on edges that come into or go out from pins and parameters nodes corresponding to the parameter.
self.base_Parameter->notEmpty() implies (
	let nodes: Set(UML::ObjectNode) =
	if self.base_Parameter.owner.oclIsKindOf(UML::Behavior) then
		let pOwner: UML::Behavior = self.base_Parameter.owner.oclAsType(UML::Behavior) in
		UML::CallBehaviorAction.allInstances()->select(a | a.behavior = pOwner)->collect(a | a.argument->at(pOwner.ownedParameter->indexOf(self.base_Parameter)))
		->union(UML::StartObjectBehaviorAction.allInstances()->select(a | a.behavior() = pOwner)->collect(a | a.argument->at(pOwner.ownedParameter->indexOf(self.base_Parameter))))
		->union(UML::ActivityParameterNode.allInstances()->select(n | n.parameter = self.base_Parameter))
		->asSet()		
	else if self.base_Parameter.owner.oclIsKindOf(UML::Operation) then
		let pOwner: UML::Operation = self.base_Parameter.owner.oclAsType(UML::Operation) in
		UML::CallOperationAction.allInstances()->select(a | a.operation = pOwner)->collect(a | a.argument->at(pOwner.ownedParameter->indexOf(self.base_Parameter)))->asSet()
	else
		Set(UML::ObjectNode){}
	endif endif in
	nodes.incoming->flatten()->union(nodes.outgoing->flatten())
	->forAll(e | let eRate: Rate = Rate.allInstances()->any(r |  r.base_ActivityEdge=e) in
		(not eRate.oclIsUndefined() and self.rate. specification.realValue() <= eRate.rate. specification.realValue()))
)
1. Allocations 
1. UML Extension
0. Stereotypes
Allocate
Constraints
[bookmark: _GoBack13]2_binary
A single «allocate» dependency shall have only one client (from) and one supplier (to).
self.base_Abstraction.source->size() = 1 and self.base_Abstraction.target->size() = 1
AllocateActivityPartition
Constraints
[bookmark: _GoBack15]1_actions_on_client_ends
An Action appearing in an "AllocateActivityPartition" shall be the /client (from) end of an "allocate" dependency. The element that represents the "AllocateActivityPartition" shall be the /supplier (to) end of the same "allocate" dependency. In the «AllocateActivityPartition» name field, Properties are designated by the use of a fully qualified name (including colon, e.g., "part_name:Block_Name"), and Classifiers are designated by a simple name (no colons, e.g., "Block_Name").
self.base_ActivityPartition.node->select(n|n.oclIsKindOf(UML::Action))
->forAll(a | let allocs: Set(UML::Abstraction) = Allocate.allInstances().base_Abstraction->select(x |x.client->includes(a))->asSet() in
	allocs->exists(x | x.supplier->includes(self.base_ActivityPartition.represents)))
[bookmark: _GoBack16]2_not_uml_semantics
The «AllocateActivityPartition» shall maintain the constraints, but not the semantics, of the UML::ActivityPartition. Classifiers or Properties represented by an «AllocateActivityPartition» do not have any direct responsibility for invoking behavior depicted within the partition boundaries. To depict this kind of direct responsibility, the modeler is directed to the UML 2 standard, sub clause 12.3.10, "ActivityPartition," Semantics topic.
-- Cannot be expressed in OCL
Requirements 
1. UML Extension
1. Stereotypes
0. [bookmark: _GoBack]Copy
Constraints
[bookmark: _GoBack128]1_source_and_taget_are_requirements
A Copy dependency may only be created between two NamedElements that have a subtype of the abstractRequirement stereotype applied
AbstractRequirement.allInstances().base_NamedElement->includesAll(self.base_Abstraction.client)
and AbstractRequirement.allInstances().base_NamedElement->includesAll(self.base_Abstraction.supplier)
[bookmark: _GoBack129]2_same_text
The text property of the client requirement is constrained to be a read-only copy of the text property of the supplier requirement and this applies recursively to all subrequirements
let cltReq: AbstractRequirement = AbstractRequirement.allInstances()->any(r | self.base_Abstraction.client->includes(r.base_NamedElement)) in
let supReq: AbstractRequirement = AbstractRequirement.allInstances()->any(r | self.base_Abstraction.supplier->includes(r.base_NamedElement)) in
self.isCopy(cltReq, supReq)
DeriveReqt
Constraints
[bookmark: _GoBack130]1_supplier_is_requirement
The supplier shall be an element stereotyped by a subtype of AbstractRequirement.
AbstractRequirement.allInstances().base_NamedElement->includesAll(self.base_Abstraction.client)
[bookmark: _GoBack131]2_client_is_requirement
The client shall be an element stereotyped by a subtype of AbstractRequirement.
AbstractRequirement.allInstances().base_NamedElement->includesAll(self.base_Abstraction.supplier)
Refine
Constraints
[bookmark: _GoBack133]2_binary
Dependencies with a Refine stereotype or one of its specializations applied shall have exactly one client and one supplier.
self.base_Abstraction.client->size()=1 and self.base_Abstraction.supplier->size()=1
Requirement
Constraints
[bookmark: _GoBack134]1_no_operation
The property "ownedOperation" shall be empty.
self.base_Class.ownedOperation->isEmpty()
[bookmark: _GoBack135]2_no_attribute
The property "ownedAttribute" shall be empty.
self.base_Class.ownedAttribute->isEmpty()
[bookmark: _GoBack136]3_no_association
Classes stereotyped by «requirement» shall not participate in associations.
UML::Association.allInstances().memberEnd->flatten().type->excludes(self.base_Class)
[bookmark: _GoBack137]4_no_generalization
Classes stereotyped by «requirement» shall not participate in generalizations.
UML::Classifier.allInstances().general->flatten()->excludes(self.base_Class)
[bookmark: _GoBack138]5_nestedclassifiers_are_requirements
A nested classifier of a class stereotyped by Requirement or one of its specializations shall also be stereotyped by Requirement or one of its specializations
self.base_Class.nestedClassifier->forAll(c | Requirement.allInstances().base_Class->includes(c))
[bookmark: _GoBack139]6_not_a_type
Classes stereotyped by «requirement» shall not be used to type any other model element.
UML::TypedElement.allInstances().type->excludes(self.base_Class)
Satisfy
Constraints
[bookmark: _GoBack140]1_supplier_is_requirement
The supplier shall be an element stereotyped by «requirement» or one of «requirement» subtypes.
AbstractRequirement.allInstances().base_NamedElement->includes(self.base_Abstraction.supplier)
TestCase
Constraints
[bookmark: _GoBack141]1_return_verdictkind
The type of return parameter of the stereotyped model element shall be VerdictKind. (note this is consistent with the UML Testing Profile).
(self.base_Behavior->notEmpty() implies self.base_Behavior.ownedParameter->exists(p | p.direction=UML::ParameterDirectionKind::return and p.type = VerdictKind ))
and
(self.base_Operation->notEmpty() implies self.base_Operation.ownedParameter->exists(p | p.direction=UML::ParameterDirectionKind::return and p.type = VerdictKind ))

Trace
Constraints
[bookmark: _GoBack143]2_binary
Dependencies with a Trace stereotype or one of its specializations applied shall have exactly one client and one supplier.
self.base_Abstraction.client->size()=1 and self.base_Abstraction.supplier->size()=1
Verify
Constraints
[bookmark: _GoBack144]1_supplier_is_requirement
The supplier shall be an element stereotyped by «requirement» or one of «requirement» subtypes.
AbstractRequirement.allInstances().base_NamedElement->includes(self.base_Abstraction.supplier)



In the specification document for each of the “Copy” stereotype definition, add the following “Operations” sub-clause after the “Constraints” sub-clause
1. Requirements 
2. UML Extension
0. Stereotypes
0. Copy
Operations
isCopy (in req1 : AbstractRequirement, in req2 : AbstractRequirement) : Boolean [1]
bodyCondition:
let subReq1: Set(AbstractRequirement) = AbstractRequirement.allInstances()->select(r | req1.base_NamedElement.ownedElement->includes(r.base_NamedElement)) in
let subReq2: Set(AbstractRequirement) = AbstractRequirement.allInstances()->select(r | req2.base_NamedElement.ownedElement->includes(r.base_NamedElement)) in
req1.text = req2.text and
subReq1->size() = subReq2->size() and
subReq1->forAll(r1 | subReq2->exists(r2 | self.isCopy(r1, r2) ))

